In[1]:=
```
NotebookDirectory[]
```

Out[1]=
```
C:\Dropbox\Work\myweb\Courses\Math_pages\Math_430\
```

Before reading, you can evaluate the entire notebook by using the menu item Evaluate -> Evaluate Notebook, shortcut Alt+v o
Since the notebook is large, it might take a few minutes for the notebook to evaluate.

Three BCs

The heat equation with the Dirichlet boundary conditions

Few solutions of the heat equation with the Dirichlet boundary conditions

Using the method of separation of variables we found “few” solutions of the heat equation with the Dirichlet boundary conditions:

$$\frac{\partial u}{\partial t} = \kappa \frac{\partial^2 u}{\partial x^2} \text{ on } \{(x, t) : 0 \le x \le L,\ t \ge 0\}$$

subject to

BCs $u(0, t) = 0,\ u(L, t) = 0$

These few solutions are

In[2]:=
```
Exp[-κ (n π/L)^2 t] * Sin[n π/L x]
```

Out[2]= $e^{-\frac{n^2 \pi^2 t \kappa}{L^2}} \text{Sin}\left[\frac{n \pi x}{L}\right]$

where **n** is any positive integer.

Verify that these functions solve the equation and satisfy BCs in *Mathematica*

In[3]:=
```
FullSimplify[
 D[Exp[-κ (n π/L)^2 t] * Sin[n π/L x], t] -
  κ D[Exp[-κ (n π/L)^2 t] * Sin[n π/L x], {x, 2}]]
```

Out[3]= 0

Or, using a more sophisticated code, called Pure Function (# is like a place holder for a function in the differential equation):

In[4]:=
```
FullSimplify[
 (D[#, t] - κ D[#, {x, 2}]) &[Exp[-κ (n π/L)^2 t] * Sin[n π/L x]]]
```

Out[4]= 0

Two boundary conditions are satisfied:

In[5]:=
```
FullSimplify[(Exp[-κ (n π/L)^2 t] * Sin[n π/L x]) /. {x → {0, L}},
 n ∈ Integers]
```

Out[5]= {0, 0}

From these few solutions we get many solutions by using the superposition principle: for arbitrary constants b_n the linear combination

$$u(x,t) = \sum_{n=1}^{nn} b_n \operatorname{Exp}\left[-\kappa \left(\frac{n\pi}{L}\right)^2 t\right] * \operatorname{Sin}\left[\frac{n\pi}{L} x\right]$$

is also a solution.

To satisfy the initial condition $u(x, 0) = f(x)$ we will need to find b_n such that

$$f(x) = \sum_{n=1}^{nn} b_n \operatorname{Sin}\left[\frac{n\pi}{L} x\right]$$

The idea is to use the orthogonality of the functions

In[6]:= $\operatorname{Sin}\left[\frac{n\pi}{L} x\right]$

Out[6]= $\operatorname{Sin}\left[\frac{n\pi x}{L}\right]$

In[7]:= $\text{FullSimplify}\left[\text{Integrate}\left[\operatorname{Sin}\left[\frac{n\,Pi}{L} x\right] \operatorname{Sin}\left[\frac{m\,Pi}{L} x\right], x\right]\right]$

Out[7]= $$\frac{L\left(\frac{\operatorname{Sin}\left[\frac{(m-n)\pi x}{L}\right]}{m-n} - \frac{\operatorname{Sin}\left[\frac{(m+n)\pi x}{L}\right]}{m+n}\right)}{2\pi}$$

In[8]:= $\text{FullSimplify}\left[\text{Integrate}\left[\operatorname{Sin}\left[\frac{n\,Pi}{L} x\right] \operatorname{Sin}\left[\frac{m\,Pi}{L} x\right], \{x, 0, L\}\right]\right]$

Out[8]= $$\frac{L\,n\operatorname{Cos}[n\pi]\operatorname{Sin}[m\pi] - L\,m\operatorname{Cos}[m\pi]\operatorname{Sin}[n\pi]}{m^2\pi - n^2\pi}$$

In[9]:=
```
FullSimplify[Integrate[Sin[n Pi/L x] Sin[m Pi/L x], {x, 0, L}],
 And[n ∈ Integers, m ∈ Integers]]
```

Out[9]= 0

The above calculation is clearly **wrong** when n=m

In[10]:=
```
FullSimplify[Integrate[Sin[n Pi/L x] Sin[n Pi/L x], {x, 0, L}],
 And[n ∈ Integers]]
```

Out[10]= $\frac{L}{2}$

The orthogonality of the first ten Sin functions is nicely seen from the table below:

In[11]:=
```
MatrixForm[
  Table[Integrate[Sin[n Pi/L x] Sin[m Pi/L x], {x, 0, L}],
   {n, 1, 10}, {m, 1, 10}]]
```

Out[11]//MatrixForm=

$$\begin{pmatrix} \frac{L}{2} & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & \frac{L}{2} & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & \frac{L}{2} & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & \frac{L}{2} & 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & \frac{L}{2} & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & \frac{L}{2} & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 & \frac{L}{2} & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 & 0 & \frac{L}{2} & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & \frac{L}{2} & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & \frac{L}{2} \end{pmatrix}$$

With $L = \pi$ we have

In[12]:=
```
Plot[
  Evaluate[Table[Sin[n Pi x / Pi], {n, 1, 10}]],
  {x, 0, Pi},
  FrameTicks → {Range[-Pi, 3 Pi, Pi/2], Range[-2, 2], {}, {}},
  Frame → True, PlotRange → {-1.1, 1.1}]
```

Out[12]=



In[13]:=
```
Manipulate[Plot[Evaluate[Sin[n x]], {x, 0, Pi},
   FrameTicks → {Range[-Pi, 3 Pi, Pi/2], Range[-2, 2], {}, {}},
   Frame → True, PlotRange → {-1.1, 1.1}],
  {{n, 1}, Range[1, 10]}, ControlType → Setter,
  ControlPlacement → Top]
```

Out[13]=



Using the orthogonality we calculate that a good candidate for an approximation of $f(x)$ is

$$f(x) \sim \sum_{n=1}^{nn} b_n \operatorname{Sin}\left[\frac{n\,\pi}{L}\,x\right]$$

where

$$b_n = \frac{2}{L}\int_0^L f(x)\,\mathrm{Sin}\left[\frac{n\,\pi}{L}\,x\right]dx$$

We will explore these approximations below.

Symbolic calculation of the solution

Example 1

In[14]:=
```
(* In this cell nn1 stands for the number of terms
 that we use to approximate a given function ff1
 which is defined by a formula on a given interval LL1 *)

Clear[nn1, ff1, LL1]; nn1 = 20;
LL1 = 1; ff1[x_] = x (x - 1)^2;
```

In[15]:=
```
Plot[Evaluate[{ff1[x]}
 ], {x, 0, LL1},
 PlotStyle → {{Blue, Thickness[0.01]},
   {Red, Thickness[0.005]}}, Frame → True,
 PlotRange → {-.02, .17}]
```

Out[15]=



Calculate the coefficients symbolically.

In[16]:=
```
Clear[bb1];
bb1[n_] = 2/LL1 Integrate[ff1[s] Sin[n Pi/LL1 s], {s, 0, LL1}]
```

Out[17]=
$$\frac{2\,(4\,n\,\pi + 2\,n\,\pi\,\text{Cos}[n\,\pi] - 6\,\text{Sin}[n\,\pi])}{n^4\,\pi^4}$$

In[18]:=
```
nn1 = 3;

(* nn1 stands for the number of terms that we use
 to approximate ff1 *)

Plot[Evaluate[{ff1[x], Sum[bb1[n] Sin[n Pi/LL1 x], {n, 1, nn1}]}
 ], {x, 0, LL1},
 PlotStyle → {{Blue, Thickness[0.01]},
   {Red, Thickness[0.005]}}, Frame → True,
 PlotRange → {-.02, .17}]
```

Out[19]=



In[20]:=
```
Manipulate[
  Plot[Evaluate[{ff1[x], Sum[bb1[n] Sin[(n Pi)/LL1 x], {n, 1, jj}]}
   ], {x, 0, LL1},
   PlotStyle → {{Blue, Thickness[0.01]},
     {Red, Thickness[0.005]}}, Frame → True,
   PlotRange → {-.02, .17}], {{jj, 5}, Range[1, 20]},
  ControlType → Setter, ControlPlacement → Top]
```

Out[20]=



In[21]:=
```
nn1 = 10;
Plot3D[
 Evaluate[
  {Sum[bb1[n] Exp[-(n Pi/LL1)^2 t] Sin[n Pi/LL1 x], {n, 1, nn1}]}
 ], {x, 0, LL1}, {t, 0, .2}, PlotRange → {-0.01, .16}]
```

Out[21]=



In[22]:=
```
Clear[t, Movie1];
Movie1[t_] :=

 Plot[
  Evaluate[
   {Sum[bb1[n] Exp[-(n Pi/LL1)^2 t] Sin[n Pi/LL1 x], {n, 1, nn1}]}
  ], {x, 0, LL1}, PlotRange → {-0.01, .16}]
```

In[24]:=
```
ListAnimate[Table[Movie1[t], {t, 0, .5, .01}],
 AnimationRunning → False, AnimationRepetitions → 2,
 ControlPlacement → Top]
```

Out[24]=



Example 2

In[25]:=
```
Clear[nn2, ff2, LL2]; nn2 = 20;
LL2 = 1; ff2[x_] = 5 √x (1 - x)^4;
```

In[26]:=
```
Plot[ff2[x], {x, 0, LL2}]
```

Out[26]=



Calculate the coefficients symbolically. If this does not work do it numerically

In[27]:=
```
Clear[bb2];

bb2[n_] =
 FullSimplify[
  2/LL2 Integrate[ff2[s] Sin[n Pi/LL2 s], {s, 0, LL2}]]
```

Out[28]=

$$\frac{1}{8\sqrt{2}\, n^{11/2}\,\pi^5} 5\left(\left(945-360\, n^2\pi^2+16\, n^4\pi^4\right)\text{FresnelC}\left[\sqrt{2}\sqrt{n}\right]+24\, n\,\pi\left(-35+4\, n^2\pi^2\right)\text{FresnelS}\left[\sqrt{2}\sqrt{n}\right]+\sqrt{2}\sqrt{n}\left(\left(-945+52\, n^2\pi^2\right)\text{Cos}[n\,\pi]+2\, n\,\pi\left(105-4\, n^2\pi^2\right)\text{Sin}[n\,\pi]\right)\right)$$

In[29]:=
```
nn2 = 50; (* increasing nn improves accuracy *)

Plot[
 Evaluate[{ff2[x], Sum[N[bb2[n]] Sin[(n Pi)/LL2 x], {n, 1, nn2}]}
 ], {x, 0, LL2},
 PlotStyle → {{Blue, Thickness[0.01]},
   {Red, Thickness[0.005]}}, Frame → True,
 PlotRange → {-.2, 1.2}]
```

Out[30]=



In[31]:=
```
Plot3D[
 Evaluate[
  {Sum[N[bb2[n]] Exp[-(n Pi/LL2)^2 t] Sin[n Pi/LL2 x], {n, 1, nn2}]}
 ], {x, 0, LL2}, {t, 0, .1}, PlotRange → {-0.01, 1.2}]
```

Out[31]=



In[32]:=
```
Clear[t, Movie2];

Movie2[t_] :=
  Plot[
   Evaluate[
    {Sum[N[bb2[n]] × Chop[Exp[-(n Pi/LL2)^2 t]] Sin[n Pi/LL2 x],
      {n, 1, nn2}]}
   ], {x, 0, LL2}, PlotRange → {-0.01, 1.2}];
```

In[34]:=
```
Movie2[0.001]
```

Out[34]=



In[35]:=
```
Off[General::munfl];
ListAnimate[Table[Movie2[t], {t, 0, 1, .01}],
 AnimationRunning → False, AnimationRepetitions → 2,
 ControlPlacement → Top]
```

Out[35]=



Example 3

In[36]:=
```
Clear[nn3, ff3, LL3]; nn3 = 20;
LL3 = Pi; ff3[x_] = UnitStep[Pi/2 - x];
```

In[37]:=
```
Plot[ff3[x], {x, 0, LL3}]
```

Out[37]=



Calculate the coefficients symbolically. If this does not work do it numerically.

In[38]:=
```
Clear[bb3];

bb3[n_] = 2/LL3 Integrate[ff3[s] Sin[n Pi/LL3 s], {s, 0, LL3}]
```

Out[39]=

$$\frac{4 \,\mathrm{Sin}\left[\frac{n \pi}{4}\right]^2}{n \pi}$$

In[40]:=
```
nn3 = 50;
Plot[
 Evaluate[{ff3[x], Sum[bb3[n] Sin[(n Pi)/LL3 x], {n, 1, nn3}]}
 ], {x, 0, LL3},
 PlotStyle → {{Blue, Thickness[0.01]},
   {Red, Thickness[0.005]}}, Frame → True,
 PlotRange → {-.2, 1.4}]
```

Out[41]=



In[42]:=
```
Plot3D[
 Evaluate[
  {Sum[bb3[n] Exp[-(n Pi/LL3)^2 t] Sin[n Pi/LL3 x], {n, 1, nn3}]}
 ], {x, 0, LL3}, {t, 0, 1}, PlotRange → {-0.2, 1.25}]
```

Out[42]=



In[43]:=
```
Clear[t, Movie3];
Movie3[t_] :=
 Plot[
  Evaluate[
   {Sum[bb3[n] Exp[-(n Pi/LL3)^2 t] Sin[n x], {n, 1, nn3}]}
  ], {x, 0, LL3}, PlotRange → {-0.2, 1.3}]
```

In[45]:=
```
ListAnimate[Table[Movie3[t], {t, 0, 1, .01}],
 AnimationRunning → False, AnimationRepetitions → 2,
 ControlPlacement → Top]
```

Out[45]=



Example 4

In[46]:=
```
Clear[nn4, ff4, LL4]; nn4 = 7;
LL4 = Pi;
ff4[x_] = Sin[x]^7;
```

Calculate the coefficients symbolically.

In[47]:=
```
Clear[bb4];
```

$$bb4[n\_] = \frac{2}{LL4} \text{Integrate}\left[ff4[s]\ \text{Sin}\left[\frac{n\ Pi}{LL4}\ s\right], \{s, 0, LL4\}\right]$$

Out[48]= $$\frac{10080\ \text{Sin}[n\ \pi]}{\left(11025 - 12916\ n^2 + 1974\ n^4 - 84\ n^6 + n^8\right)\ \pi}$$

There are problems with this formula!

In[49]:=
```
bb4[3]
```

Power: Infinite expression $\frac{1}{0}$ encountered.

Infinity: Indeterminate expression $\frac{0\ \text{ComplexInfinity}}{\pi}$ encountered.

Out[49]= Indeterminate

In[50]:=
```
Factor[Denominator[bb4[n]]]
```

Out[50]= $(-7+n)\ (-5+n)\ (-3+n)\ (-1+n)\ (1+n)\ (3+n)\ (5+n)\ (7+n)\ \pi$

In[51]:=
```
bbl4 = Table[Limit[bb4[n], n → k], {k, 1, 12, 1}]
```

Out[51]= $\left\{\frac{35}{64}, 0, -\frac{21}{64}, 0, \frac{7}{64}, 0, -\frac{1}{64}, 0, 0, 0, 0, 0\right\}$

The same numbers would have been obtained if we calculate

In[52]:= $$\text{Table}\left[\frac{2}{LL4} \text{Integrate}\left[ff4[s]\ \text{Sin}\left[\frac{n\ Pi}{LL4}\ s\right], \{s, 0, LL4\}\right], \{n, 1, 12\}\right]$$

Out[52]= $\left\{\frac{35}{64}, 0, -\frac{21}{64}, 0, \frac{7}{64}, 0, -\frac{1}{64}, 0, 0, 0, 0, 0\right\}$

Thus in this case we have the exact solution if we use seven terms in the expansion of ff[x].

In[53]:=
```
nn4 = 7;
Plot[
 Evaluate[{ff4[x], Sum[bbl4[[n]] Sin[(n Pi)/LL4 x], {n, 1, nn4}]}
 ], {x, 0, LL4},
 PlotStyle → {{Blue, Thickness[0.01]},
   {Red, Thickness[0.005]}}, Frame → True,
 PlotRange → {-.2, 1.1}]
```

Out[54]=



In fact we have the exact formula for ff in this case

In[55]:=
```
FullSimplify[ff4[x] == Sum[bbl4[[n]] Sin[(n Pi)/LL4 x], {n, 1, nn4}]]
```

Out[55]= True

In[56]:=
```
Plot3D[
 Evaluate[
  {Sum[bbl4[[n]] Exp[-(n Pi/LL4)^2 t] Sin[n Pi/LL4 x], {n, 1, nn4}]}
 ], {x, 0, LL4}, {t, 0, 1}, PlotRange → {-0.2, 1.1}]
```

Out[56]=



In[57]:=
```
Clear[t, Movie4];
Movie4[t_] :=
 Plot[
  Evaluate[
   {Sum[bbl4[[n]] Exp[-(n Pi/LL4)^2 t] Sin[n x], {n, 1, nn4}]}
  ], {x, 0, LL4}, PlotRange → {-0.1, 1.1}]
```

In[59]:=
```
ListAnimate[Table[Movie4[t], {t, 0, 1, .01}],
 AnimationRunning → False, AnimationRepetitions → 2,
 ControlPlacement → Top]
```

Out[59]=



Numeric calculation of an approximation for the solution

Example 5

In[60]:=
```
Clear[mm5, gg5, LL5]; mm5 = 20;
LL5 = Pi; gg5[x_] = Sin[x]^(1/2);
```

Calculate the coefficients symbolically. (Not working.)

In[61]:=
```
(* bb[n_]=
  2/LL Integrate[gg[s] Sin[n Pi/LL s],{s,0,LL},
   Assumptions→And[n∈Integers,n>0]]
*)
```

In[62]:=
```
Clear[bbl5];

bbl5 = Chop[Table[2/LL5 NIntegrate[
     gg5[s] Sin[n Pi/LL5 s],
     {s, 0, LL5},
     Method → {Automatic}, MaxRecursion → 200,
     AccuracyGoal → 12, PrecisionGoal → 16], {n, 1, mm5}]]
```

Out[63]=
```
{1.11284, 0, 0.158977, 0, 0.0722621, 0,
 0.0433572, 0, 0.0296655, 0, 0.0219267, 0, 0.0170541,
 0, 0.0137533, 0, 0.0113956, 0, 0.00964241, 0}
```

In[64]:=
```
Plot[
  Evaluate[{gg5[x], Sum[bbl5[[n]] Sin[(n Pi)/LL5 x], {n, 1, mm5}]}
  ], {x, 0, LL5},
  PlotStyle → {{Blue, Thickness[0.01]},
    {Red, Thickness[0.005]}}, Frame → True,
  PlotRange → {-.2, 1.15}]
```

Out[64]=



In[65]:=
```
Plot3D[
  Evaluate[Sum[bbl5[[n]] Exp[-(n Pi/LL5)^2 t] Sin[n Pi/LL5 x],
    {n, 1, mm5}]
  ], {x, 0, LL5}, {t, 0, 2}, PlotRange → {-0.1, 1.1}]
```

Out[65]=



In[66]:=
```
Clear[t, Movie5];

Movie5[t_] :=
 Plot[Evaluate[Sum[bbl5[[n]] Exp[-(n Pi/LL5)^2 t] Sin[n Pi/LL5 x],
     {n, 1, mm5}]
   ], {x, 0, LL5}, PlotRange → {-0.1, 1.1}]
```

In[68]:=
```
ListAnimate[Table[Movie5[t], {t, 0, 2, .05}],
 AnimationRunning → False, AnimationRepetitions → 2,
 ControlPlacement → Top]
```

Out[68]=



Example 6

In[69]:=
```
Clear[mm6, gg6, LL6]; mm6 = 20;
LL6 = Pi; gg6[x_] = Sin[x]^(1/10);
```

Calculate the coefficients numerically.

In[70]:=
```
Clear[bbl6];

bbl6 =
 Chop[
  Table[2/LL6 NIntegrate[gg6[s] Sin[n Pi/LL6 s], {s, 0, LL6},
     Method → {Automatic}, MaxRecursion → 200,
     AccuracyGoal → 12, PrecisionGoal → 16], {n, 1, mm6, 1}]]
```

Out[71]=
```
{1.23582, 0, 0.358787, 0, 0.204016, 0,
 0.1408, 0, 0.10676, 0, 0.0856006, 0, 0.0712249,
 0, 0.0608478, 0, 0.0530194, 0, 0.0469124, 0}
```

In[72]:=
```
Plot[
  Evaluate[{gg6[x], Sum[bbl6[[n]] Sin[(n Pi)/LL6 x], {n, 1, mm6}]}
  ], {x, -0., LL6},
  PlotStyle → {{Blue, Thickness[0.01]},
    {Red, Thickness[0.005]}}, Frame → True,
  PlotRange → {{-0.1, LL6 + .1}, {-.2, 1.25}},
  Exclusions → None, PlotPoints → 400]
```

Out[72]=



In[73]:=
```
Plot3D[
 Evaluate[
  {Sum[bbl6[[n]] Exp[-(n Pi/LL6)^2 t] Sin[n Pi/LL6 x], {n, 1, mm6}]}
 ], {x, 0, LL6}, {t, 0, 2}, PlotRange → {-0.1, 1.15},
 PlotPoints → 200]
```

Out[73]=



In[74]:=
```
Clear[t, Movie6];
Movie6[t_] :=
 Plot[
  Evaluate[
   {Sum[bbl6[[n]] Exp[-(n Pi/LL6)^2 t] Sin[n Pi/LL6 x], {n, 1, mm6}]}
  ], {x, 0, LL6}, PlotRange → {-0.1, 1.1}]
```

In[76]:=
```
ListAnimate[Table[Movie6[t], {t, 0, 2, .05}],
  AnimationRunning → False, AnimationRepetitions → 2,
  ControlPlacement → Top]
```

Out[76]=



The heat equation with the Neumann boundary conditions

Few solutions of the heat equation with the Neumann boundary conditions

Using the method of separation of variables we found “few” solutions of the heat equation with the Neumann boundary conditions:

$$\frac{\partial u}{\partial t} = \kappa \frac{\partial^2 u}{\partial x^2} \text{ on } \{(x, t) : 0 \le x \le L,\ t \ge 0\}$$

subject to

BCs $\frac{\partial u}{\partial x}(0, t) = 0$, $\frac{\partial u}{\partial x}(L, t) = 0$

These few solutions are

In[77]:= **1**

Out[77]= 1

and

In[78]:= **Exp[-κ (nπ/L)^2 t] * Cos[nπ/L x]**

Out[78]= $e^{-\frac{n^2 \pi^2 t \kappa}{L^2}} \text{Cos}\left[\frac{n \pi x}{L}\right]$

where **n** is any positive integer. From these few soluting we get many solutions by using the superposition principle: for arbitrary constants a_n the linear combination

$$u(x, t) = a_0 + \sum_{n=1}^{nn} a_n \text{Exp}\left[-\kappa \left(\frac{n \pi}{L}\right)^2 t\right] * \text{Cos}\left[\frac{n \pi}{L} x\right]$$

is also a solution.

To satisfy the initial condition $u(x, 0) = f(x)$ we will need to find a_n such that

$$f(x) = a_0 + \sum_{n=1}^{nn} a_n \text{Cos}\left[\frac{n \pi}{L} x\right]$$

The idea is to use the orthogonality of the functions

In[79]:=
```
Cos[n π/L x]
```

Out[79]= $\text{Cos}\left[\frac{n\,\pi\,x}{L}\right]$

In[80]:=
```
Clear[L];
FullSimplify[Integrate[Cos[n Pi/L x] Cos[m Pi/L x], x]]
```

Out[80]= $\frac{L\left(\frac{\text{Sin}\left[\frac{(m-n)\,\pi\,x}{L}\right]}{m-n}+\frac{\text{Sin}\left[\frac{(m+n)\,\pi\,x}{L}\right]}{m+n}\right)}{2\,\pi}$

In[81]:=
```
Clear[L];
FullSimplify[Integrate[Cos[n Pi/L x] Cos[m Pi/L x], {x, 0, L}]]
```

Out[81]= $\frac{L\,m\,\text{Cos}[n\,\pi]\,\text{Sin}[m\,\pi]-L\,n\,\text{Cos}[m\,\pi]\,\text{Sin}[n\,\pi]}{m^2\,\pi-n^2\,\pi}$

In[82]:=
```
Clear[L];
FullSimplify[Integrate[Cos[n Pi/L x] Cos[m Pi/L x], {x, 0, L}],
 And[n ∈ Integers, m ∈ Integers]]
```

Out[82]= 0

The above calculation is clearly wrong when n=m

In[83]:=
```
Clear[L];
FullSimplify[Integrate[Cos[n Pi/L x] Cos[n Pi/L x], {x, 0, L}],
 And[n ∈ Integers]]
```

Out[83]= $\frac{L}{2}$

The orthogonality of the first ten Cos functions and the constant is nicely seen from the table below:

In[84]:=
```
MatrixForm[
 Table[Integrate[Cos[n Pi/L x] Cos[m Pi/L x], {x, 0, L}],
  {n, 0, 10}, {m, 0, 10}]]
```

Out[84]//MatrixForm=

$$\begin{pmatrix} L & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & \frac{L}{2} & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & \frac{L}{2} & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & \frac{L}{2} & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & \frac{L}{2} & 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & \frac{L}{2} & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 & \frac{L}{2} & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 & 0 & \frac{L}{2} & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & \frac{L}{2} & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & \frac{L}{2} & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & \frac{L}{2} \end{pmatrix}$$

With $L = \pi$ we have

In[85]:=
```
Plot[
  Evaluate[Table[Cos[n Pi x / Pi], {n, 0, 10}]],
  {x, 0, Pi},
  FrameTicks → {Range[-Pi, 3 Pi, Pi/2], Range[-2, 2], {}, {}},
  Frame → True, PlotRange → {-1.1, 1.1}]
```

Out[85]=



In[86]:=
```
Manipulate[Plot[Evaluate[Cos[n x]], {x, 0, Pi},
  FrameTicks → {Range[-Pi, 3 Pi, Pi/2], Range[-2, 2], {}, {}},
  Frame → True, PlotRange → {-1.1, 1.1}],
 {{n, 1}, Range[0, 10]}, ControlType → Setter,
 ControlPlacement → Top]
```

Out[86]=



Using the orthogonality we calculate that a good candidate for approximation of $f(x)$ is

$$f(x) \sim a_0 + \sum_{n=1}^{nn} a_n \operatorname{Cos}\left[\frac{n\pi}{L} x\right]$$

where

$$a_0 = \frac{1}{L}\int_0^L f(x)\,dx,\ \ a_n = \frac{2}{L}\int_0^L f(x)\,\text{Cos}\left[\frac{n\,\pi}{L}x\right]dx,\ n > 0.$$

We will explore these approximations below.

Symbolic calculation of the solution

Example 7

In[87]:=
```
Clear[nn7, ff7, LL7]; nn7 = 20;
LL7 = 1; ff7[x_] = x^2 (x - 2)^2;
```

In[88]:=
```
Plot[Evaluate[ff7[x]], {x, 0, LL7},
 PlotStyle → {{Blue, Thickness[0.01]},
   {Red, Thickness[0.005]}}, Frame → True,
 PlotRange → {-.02, 1.2}]
```

Out[88]=



Calculate the coefficients symbolically.

In[89]:=
```
Clear[aa7];

aa7[0] = 1/LL7 Integrate[ff7[s], {s, 0, LL7}];

aa7[n_] = 2/LL7 Integrate[ff7[s] Cos[n Pi/LL7 s], {s, 0, LL7}]
```

Out[91]= $\frac{2\left(-24\, n\, \pi+\left(24+4\, n^{2}\, \pi^{2}+n^{4}\, \pi^{4}\right) \operatorname{Sin}[n\, \pi]\right)}{n^{5}\, \pi^{5}}$

In[92]:=
```
nn7 = 7;

(* nn7 stands for the number of terms that we use
 to approximate ff *)

Plot[Evaluate[{ff7[x], Sum[aa7[n] Cos[(n Pi)/(LL7) x], {n, 0, nn7}]}
 ], {x, 0, LL7},
 PlotStyle → {{Blue, Thickness[0.01]},
   {Red, Thickness[0.005]}}, Frame → True,
 PlotRange → {-.02, 1.2}]
```

Out[93]=



In[94]:=
```
Plot3D[
 Evaluate[
  {Sum[aa7[n] Exp[-(n Pi/LL7)^2 t] Cos[n Pi/LL7 x], {n, 0, nn7}]}
 ], {x, 0, LL7}, {t, 0, .2}, PlotRange → {-0.01, 1.2}]
```

Out[94]=



In[95]:=
```
Clear[t, Movie7];
Movie7[t_] :=

 Plot[
  Evaluate[
   {Sum[aa7[n] Exp[-(n Pi/LL7)^2 t] Cos[n Pi/LL7 x], {n, 0, nn7}]}
  ], {x, 0, LL7}, PlotRange → {-0.01, 1.2}]
```

In[97]:=
```
ListAnimate[Table[Movie7[t], {t, 0, .2, .005}],
 AnimationRunning → False, AnimationRepetitions → 2,
 ControlPlacement → Top]
```

Out[97]=



Numeric calculation of the solution

Example 8

In[98]:=
```
Clear[mm8, gg8, LL8]; mm8 = 20;
LL8 = Pi; gg8[x_] = (Cos[x / 2]^8)^(1/3);
```

In[99]:=
```
Plot[Evaluate[{gg8[x]}
 ], {x, 0, LL8},
 PlotStyle → {{Blue, Thickness[0.01]},
   {Red, Thickness[0.005]}}, Frame → True,
 PlotRange → {-.2, 1.15}]
```

Out[99]=



In[100]:=
```
Clear[aal8];

aal8 = Chop[Prepend[Table[2/LL8 NIntegrate[
     gg8[s] Cos[n Pi/LL8 s],
     {s, 0, LL8},
     Method → {Automatic}, MaxRecursion → 200,
     AccuracyGoal → 12, PrecisionGoal → 16], {n, 1, mm8}],
   1/LL8 NIntegrate[
    gg8[s],
    {s, 0, LL8},
    Method → {Automatic}, MaxRecursion → 200,
    AccuracyGoal → 12, PrecisionGoal → 16]]]
```

Out[101]= {0.445734, 0.50941, 0.050941, -0.00783708, 0.00244909, -0.00103119, 0.000515597, -0.000288734, 0.000175303, -0.000113099, 0.000076508, -0.0000537624, 0.0000389777, -0.0000290067, 0.0000220703, -0.0000171157, 0.0000134951, -0.0000107961, 8.74855×10^{-6}, -7.17094×10^{-6}, 5.93844×10^{-6}}

In[102]:=
```
Length[aal8]
```

Out[102]= 21

In[103]:=
```
Plot[
 Evaluate[
  {gg8[x], Sum[aal8[[n + 1]] Cos[(n Pi)/LL8 x],
    {n, 0, Length[aal8] - 1}]}
 ], {x, 0, LL8},
 PlotStyle → {{Blue, Thickness[0.01]},
   {Red, Thickness[0.005]}}, Frame → True,
 PlotRange → {-.2, 1.15}]
```

Out[103]=



In[104]:=
```
Plot3D[
  Evaluate[
    {Sum[aal8[[n + 1]] Exp[-(n Pi/LL8)^2 t] Cos[n Pi/LL8 x],
      {n, 0, Length[aal8] - 1}]}
  ], {x, 0, LL8}, {t, 0, 3}, PlotRange → {-0.1, 1.1}]
```

Out[104]=



In[105]:=
```
Clear[t, Movie8];

Movie8[t_] :=
 Plot[
  Evaluate[
   {Sum[aal8[[n + 1]] Exp[-(n Pi/LL8)^2 t] Cos[n Pi/LL8 x],
     {n, 0, Length[aal8] - 1}]}
  ], {x, 0, LL8}, PlotRange → {-0.1, 1.1}]
```

In[107]:=
```
ListAnimate[Table[Movie8[t], {t, 0, 3, .05}],
 AnimationRunning → False, AnimationRepetitions → 2,
 ControlPlacement → Top]
```

Out[107]=



The heat equation with the periodic

boundary conditions

Few solutions of the heat equation with the periodic boundary conditions

Using the method of separation of variables we found “few” solutions of the heat equation with the periodic boundary conditions:

$$\frac{\partial u}{\partial t} = \kappa \frac{\partial^2 u}{\partial x^2} \text{ on } \{(x, t) : -L \le x \le L,\ t \ge 0\}$$

subject to

BCs: $u(-L, t) = u(L, t)$, $\frac{\partial u}{\partial x}(-L, t) = \frac{\partial u}{\partial x}(L, t)$

These few solutions are

```
In[108]:= 1
Out[108]= 1
```

and

```
In[109]:= Exp[-κ (n π / L)^2 t] * Cos[n π / L x]
```

$$\text{Out[109]=}\quad e^{-\frac{n^2 \pi^2 t \kappa}{L^2}} \operatorname{Cos}\left[\frac{n \pi x}{L}\right]$$

and

In[110]:= $\text{Exp}\left[-\kappa\left(\frac{n\,\pi}{L}\right)^2 t\right] * \text{Sin}\left[\frac{n\,\pi}{L}\,x\right]$

Out[110]= $e^{-\frac{n^2 \pi^2 t\,\kappa}{L^2}}\,\text{Sin}\left[\frac{n\,\pi\,x}{L}\right]$

where **n** is any positive integer. From these few soluting we get many solutions by using the superposition principle: for arbitrary constants b_n the linear combination

$$u(x, t) = a_0 + \sum_{n=1}^{nn} a_n\,\text{Exp}\left[-\kappa\left(\frac{n\,\pi}{L}\right)^2 t\right] * \text{Cos}\left[\frac{n\,\pi}{L}\,x\right] + \sum_{n=1}^{nn} b_n\,\text{Exp}\left[-\kappa\left(\frac{n\,\pi}{L}\right)^2 t\right] * \text{Sin}\left[\frac{n\,\pi}{L}\,x\right]$$

is also a solution.

To satisfy the initial condition $u(x, 0) = f(x)$ we will need to find a_ns and b_ns such that

$$f(x) = a_0 + \sum_{n=1}^{nn} a_n\,\text{Cos}\left[\frac{n\,\pi}{L}\,x\right] + \sum_{n=1}^{nn} b_n\,\text{Sin}\left[\frac{n\,\pi}{L}\,x\right]$$

The idea is to use the orthogonality of the functions

In[111]:= **1**

Out[111]= 1

and

In[112]:= $\text{Cos}\left[\frac{n\,\pi}{L}\,x\right]$

Out[112]= $\text{Cos}\left[\frac{n\,\pi\,x}{L}\right]$

and

In[113]:=
```
Sin[n π/L x]
```

Out[113]= $\text{Sin}\left[\frac{n\,\pi\,x}{L}\right]$

on the interval [-L,L].

In[114]:=
```
Clear[L];
FullSimplify[Integrate[Cos[n Pi/L x] Cos[m Pi/L x], {x, -L, L}]]
```

Out[114]= $\frac{L\left(\frac{\text{Sin}[(m-n)\,\pi]}{m-n}+\frac{\text{Sin}[(m+n)\,\pi]}{m+n}\right)}{\pi}$

In[115]:=
```
Clear[L];
FullSimplify[Integrate[Cos[n Pi/L x] Cos[m Pi/L x], {x, -L, L}],
 And[n ∈ Integers, m ∈ Integers]]
```

Out[115]= 0

The above calculation is clearly wrong when n=m

In[116]:=
```
Clear[L];
FullSimplify[Integrate[Cos[n Pi/L x] Cos[n Pi/L x], {x, -L, L}],
 And[n ∈ Integers]]
```

Out[116]= L

Similarly

In[117]:=
```
Clear[L];
FullSimplify[Integrate[Sin[n Pi/L x] Sin[m Pi/L x], {x, -L, L}],
 And[n ∈ Integers, m ∈ Integers]]
```

Out[117]= 0

The above calculation is clearly wrong when n=m

In[118]:=
```
Clear[L];
FullSimplify[Integrate[Sin[n Pi/L x] Sin[n Pi/L x], {x, -L, L}],
 And[n ∈ Integers]]
```

Out[118]= L

and finally

In[119]:=
```
Clear[L];
FullSimplify[Integrate[Cos[n Pi/L x] Sin[m Pi/L x], {x, -L, L}],
 And[n ∈ Integers, m ∈ Integers]]
```

Out[119]= 0

The orthogonality of the first ten Cos functions and the constant is nicely seen from the table below:

In[120]:=
```
CoSiFunT = Join[Table[Cos[n π/L x], {n, 0, 5}],
  Table[Sin[n π/L x], {n, 1, 5}]]
```

Out[120]= $\left\{1, \text{Cos}\left[\frac{\pi x}{L}\right], \text{Cos}\left[\frac{2 \pi x}{L}\right], \text{Cos}\left[\frac{3 \pi x}{L}\right], \text{Cos}\left[\frac{4 \pi x}{L}\right], \text{Cos}\left[\frac{5 \pi x}{L}\right], \text{Sin}\left[\frac{\pi x}{L}\right], \text{Sin}\left[\frac{2 \pi x}{L}\right], \text{Sin}\left[\frac{3 \pi x}{L}\right], \text{Sin}\left[\frac{4 \pi x}{L}\right], \text{Sin}\left[\frac{5 \pi x}{L}\right]\right\}$

In[121]:=
```
MatrixForm[
 Table[Integrate[CoSiFunT[[j]] CoSiFunT[[k]], {x, -L, L}],
  {j, 1, Length[CoSiFunT]}, {k, 1, Length[CoSiFunT]}]
]
```

Out[121]//MatrixForm=

$$\begin{pmatrix} 2L & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & L & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & L & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & L & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & L & 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & L & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 & L & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 & 0 & L & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & L & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & L & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & L \end{pmatrix}$$

Using the orthogonality we calculate that a good candidate for approximation of $f(x)$ is

$$f(x) \sim a_0 + \sum_{n=1}^{nn} a_n \operatorname{Cos}\left[\frac{n\pi}{L} x\right] + \sum_{n=1}^{nn} b_n \operatorname{Sin}\left[\frac{n\pi}{L} x\right]$$

where

$$a_0 = \frac{1}{2L} \int_{-L}^{L} f(x)\, dx, \quad a_n = \frac{1}{L} \int_0^L f(x) \operatorname{Cos}\left[\frac{n\pi}{L} x\right] dx,$$

$$b_n = \frac{1}{L} \int_{-L}^{L} f(x) \operatorname{Sin}\left[\frac{n\pi}{L} x\right] dx, \; n > 0.$$

We will explore these approximations below.

Symbolic calculation of the solution

Example 9

In[122]:=
```
Clear[nn9, ff9, LL9]; nn9 = 20;
LL9 = 1; ff9[x_] = (x^2 - 1) (3 x - 1) + 1;
```

In[123]:=
```
Plot[Evaluate[{ff9[x]}
  ], {x, -LL9, LL9},
  PlotStyle → {{Blue, Thickness[0.01]},
    {Red, Thickness[0.005]}}, Frame → True,
  PlotRange → {0, 3}]
```

Out[123]=



Calculate the coefficients symbolically.

In[124]:=
```
Clear[aa9];

aa9[0] = 1/(2 LL9) Integrate[ff9[s], {s, -LL9, LL9}];

aa9[n_] = 1/LL9 Integrate[ff9[s] Cos[n Pi/LL9 s], {s, -LL9, LL9}];

bb9[n_] = 1/LL9 Integrate[ff9[s] Sin[n Pi/LL9 s], {s, -LL9, LL9}]
```

Out[127]=
$$\frac{12\left(3 n \pi \operatorname{Cos}[n \pi]+\left(-3+n^2 \pi^2\right) \operatorname{Sin}[n \pi]\right)}{n^4 \pi^4}$$

In[128]:=
```
nn9 = 7;

(* nn stands for the number of terms that we use
 to approximate ff *)

Plot[
 Evaluate[
  {ff9[x], Sum[aa9[n] Cos[(n Pi)/LL9 x], {n, 0, nn9}] +
    Sum[bb9[n] Sin[(n Pi)/LL9 x], {n, 1, nn9}]}
 ], {x, -LL9, LL9},
 PlotStyle → {{Blue, Thickness[0.01]},
   {Red, Thickness[0.005]}}, Frame → True,
 PlotRange → {0, 3}]
```

Out[129]=



In[130]:=
```
Plot3D[
 Evaluate[
  {Sum[aa9[n] Exp[-(n Pi/LL9)^2 t] Cos[n Pi/LL9 x], {n, 0, nn9}] +
    Sum[bb9[n] Exp[-(n Pi/LL9)^2 t] Sin[n Pi/LL9 x], {n, 1, nn9}]}
 ], {x, -LL9, LL9}, {t, 0, .5}, PlotRange → {0, 3}]
```

Out[130]=



In[131]:=
```
Clear[t, Movie9];
Movie9[t_] :=

 Plot[
  Evaluate[
   {Sum[aa9[n] Exp[-(n Pi/LL9)^2 t] Cos[n Pi/LL9 x], {n, 0, nn9}] +
     Sum[bb9[n] Exp[-(n Pi/LL9)^2 t] Sin[n Pi/LL9 x], {n, 1, nn9}]}
  ], {x, -LL9, LL9}, PlotRange → {0, 3}]
```

In[133]:=
```
ListAnimate[Table[Movie9[t], {t, 0, 1, .005}],
 AnimationRunning → False, AnimationRepetitions → 2,
 ControlPlacement → Top]
```

Out[133]=



Numeric calculation of the solution

Example 10

In[134]:=
```
Clear[mm10, gg10, LL10]; mm10 = 10;
LL10 = 1; gg10[x_] = ((-Sinh[1] x + Cosh[1]) Exp[x])^(3/2);
```

In[135]:=
```
gg10[-1] == gg10[1]
```

Out[135]=
```
True
```

In[136]:=
```
Plot[Evaluate[{gg10[x]}
  ], {x, -LL10, LL10},
 PlotStyle → {{Blue, Thickness[0.01]},
   {Red, Thickness[0.005]}}, Frame → True,
 PlotRange → {0, 2.3}]
```

Out[136]=



Try symbolic integration (slow)

```
In[137]:= (* 1/LL Integrate[
    gg[s] Cos[n Pi/LL s],
    {s,-LL,LL}] *)

In[138]:= Clear[aal10];

aal10 = Chop[Prepend[Table[1/LL10 NIntegrate[
      gg10[s] Cos[n Pi/LL10 s],
      {s, -LL10, LL10},
      Method → {Automatic}, MaxRecursion → 200,
      AccuracyGoal → 12, PrecisionGoal → 16], {n, 1, mm10}],
   1/(2 LL10) NIntegrate[
    gg10[s],
    {s, -LL10, LL10},
    Method → {Automatic}, MaxRecursion → 200,
    AccuracyGoal → 12, PrecisionGoal → 16]]]

Out[139]= {1.6345, 0.358735, -0.107704, 0.0478621,
  -0.0267352, 0.0170227, -0.0117805, 0.00863474,
  -0.00660008, 0.00520868, -0.0042153}

In[140]:= Length[aal10]

Out[140]= 11
```

In[141]:=
```
bbl10 = Chop[Table[1/LL10 NIntegrate[
      gg10[s] Sin[n Pi/LL10 s],
      {s, -LL10, LL10},
      Method → {Automatic}, MaxRecursion → 200,
      AccuracyGoal → 12, PrecisionGoal → 16], {n, 1, mm10}]]
```

Out[141]=
```
{0.260404, -0.0285929, 0.00743374,
 -0.00290449, 0.0014201, -0.000798147, 0.000492931,
 -0.000325787, 0.000226592, -0.000163996}
```

In[142]:=
```
Plot[
 Evaluate[
  {gg10[x],
   Sum[aal10[[n + 1]] Cos[(n Pi)/LL10 x], {n, 0, Length[aal10] - 1}] +
    Sum[bbl10[[n]] Sin[(n Pi)/LL10 x], {n, 1, Length[bbl10]}]}
  ], {x, -LL10, LL10},
 PlotStyle -> {{Blue, Thickness[0.01]},
   {Red, Thickness[0.005]}}, Frame -> True,
 PlotRange -> {0, 2.3}]
```

Out[142]=



In[143]:=
```
Plot3D[
 Evaluate[
  {Sum[aal10[[n + 1]] Exp[-(n Pi/LL10)^2 t] Cos[n Pi/LL10 x],
     {n, 0, Length[aal10] - 1}] +
   Sum[bbl10[[n]] Exp[-(n Pi/LL10)^2 t] Sin[n Pi/LL10 x],
     {n, 1, Length[bbl10]}]}
 ], {x, -LL10, LL10}, {t, 0, .2}, PlotRange → {0.8, 2.2}]
```

Out[143]=



In[144]:=
```
Clear[t, Movie10];

Movie10[t_] :=
 Plot[
  Evaluate[
   {Sum[aal10[[n + 1]] Exp[-(n Pi/LL10)^2 t] Cos[n Pi/LL10 x],
      {n, 0, Length[aal10] - 1}] +
     Sum[bbl10[[n]] Exp[-(n Pi/LL10)^2 t] Sin[n Pi/LL10 x],
      {n, 1, Length[bbl10]}]}
   ], {x, -LL10, LL10}, PlotRange → {0.8, 2.2}]
```

In[146]:=
```
ListAnimate[Table[Movie10[t], {t, 0, .5, .01}],
 AnimationRunning → False, AnimationRepetitions → 2,
 ControlPlacement → Top]
```

Out[146]=